
JMASM Algorithms and Code
JMASM10: A Fortran Routine For Sieve Bootstrap Prediction Intervals
Andrés M. Alonso
Department of Mathematics
Universidad Autónoma de Madrid
A Fortran routine for constructing nonparametric prediction intervals for a general class of linear
processes is described. The approach uses the sieve bootstrap procedure of Bühlmann (1997) based on
residual resampling from an autoregressive approximation to the given process.
Key words: Sieve bootstrap, prediction, time series
also include the parameter estimation variability.
This article describes a bootstrap
method to construct nonparametric prediction
intervals for a class of linear processes that can
be written as a one-sided infinite-order moving
average process with at most a polynomial decay

Introduction
When studying a time series, one of the goals is
the estimation of forecast confidence intervals
based on an observed trajectory of the process.
The traditional approach of finding prediction
intervals for a linear time series assumes that the
distribution of the error process is known. Thus,
these prediction intervals could be adversely
affected by departures from the true underlying
distribution.
Some bootstrap approaches have been
proposed as a distribution free alternative to
compute prediction intervals. Stine (1987)
proposed a bootstrap method to estimate the
prediction mean squared error of the estimated
linear predictor of an AR(p) where p is known.
Also, for an AR(p) process with known p, and
relaxing the assumptions of Stine (1987),
Thombs and Schucany (1990) propose a
backward then forward bootstrap method to
estimate prediction intervals. Cao et al. (1997)
study a conditional bootstrap method alternative
to Thombs and Schucany's proposal, which is
computationally faster. Pascual et al. (2001)
generalize this conditional bootstrap to
ARMA(p, q) processes with known p and q and

{ }

of the coefficients ψ

+∞

j

j =0

. This class includes

the stationary and invertible ARMA(p,q)
processes. This approach uses the sieve
bootstrap of Bühlmann (1997) based on residual
resampling from a sequence of approximating
autoregressions for {X t }t∈Z with order p = p(n)
that increases as a function of the sample size n.
This sieve bootstrap has a nice
nonparametric property, being model-free within
the considered class of linear processes. Thus,
the proposed bootstrap prediction intervals could
be applied to this more general class of linear
models without specifying a finite dimensional
model as in previous bootstrap proposals.
Alonso et al. (2002) and (2003) studied the
consistency and the finite sample properties of
this sieve bootstrap.
Methodology
Let {X t }t∈Z be a real valued, stationary process

with expectation E [X t ] = µ X that admits a

MA(∞) representation with

∑

+∞

ψ 2j < ∞ .

j =0

Under the additional assumption of invertibility
{X t }t∈Z can be represented as a one-sided
infinite-order autoregressive process:

∑ j =0 φ j ( X t − j − µ X ) = ε t ,
+∞

5.

φ 0 = 1, t ∈ Z
(1)

with

{φ }

+∞

coefficients

j

Draw

ε t*

resample

of

i.i.d.

observations from F̂ε~ .
Define X t* by the recursion:

6.

∑

satisfying

j =0

a

p
j =0

(4)

φ j ( X t*− j − X ) = ε t* ,

φ j2 < ∞ . This AR(∞) representation

where the starting p observations are equal to

motivates Bühlmann's sieve bootstrap. The
method proceeds as follows:

In practice an AR(p) resample is
generated using (4) with sample size equal to n
+ 100 and then discard the first 100
observations. Up to this step, the resampling
plan coincides with the sieve bootstrap, and is
valid for bootstrapping some statistics defined as
a functional of a m-dimensional distribution
function (see details in Section 3.3 of Bühlmann,
1997). However, it is not effective for bootstrap
prediction, because it does not replicate the
conditional distribution of X n + h given the
observed data. But, proceeding as do Cao et al.
(1997) by fixing the last p observations
resamples of the future values can be obtained
X n*+ h
given
X n*− p +1 = X n − p +1 ,

∑

+∞
j =0

1.
Given a sample {X 1 , X 2 , … , X n },
select the order p = p(n) of the autoregressive
approximation by AICC criterion: AICC =
− n log(σ 2 ) + 2( p + 1)n /(n − p − 2) ,
(cf.
Section 9.3 of Brockwell &Davis, 1991).
The AICC criterion is a bias-corrected
version of AIC (Akaike, 1973), and it has a more
extreme penalty for large-order models which
counteracts the overfitting nature of AIC. Other
order selection criteria (such as BIC) could be
used, but AICC is preferred assuming the view
that the true model is complex and not of finite
dimension, and also because the AICC is
asymptotically efficient for autoregressive
models, i.e., it chooses an AR model which
achieves the optimal rate of convergence of the
mean-square prediction error.
2.
Construct some estimators of the
autoregressive

(φˆ ,φˆ ,…,φˆ ) .

coefficients:

1

2

p

Following Bühlmann (1997) the Yule-Walker
estimates are taken.
3.
Compute the residuals for t ∈ (p+1,p+2,
…, n) by:

εˆt = ∑ j =0 φˆ j ( X t − j − X ),

4.
Define the empirical
function of the centered residuals:

X n*− p + 2 = X n − p + 2 , … , X n* = X n .
7.

Compute

the

estimation

(

of

autoregressive coefficients: φˆ , φˆ , … , φˆ
*
1

*
2

in step 1.
8.
Compute
the
future
observations by the recursion:

)

the
as

bootstrap

X n*+ h = X − ∑ j =1φˆ *j ( X t*− j − X ) + ε t* ,
p

*
p

(5)

where h > 0, and X t* = X t , for t ≤ n.
Finally,

FX* * ( x)
n+h

the

bootstrap

(2)

distribution of X n*+ h is used to approximate the

distribution

unknown distribution of X n + h given the
observed sample. As usual, a Monte Carlo

φˆ0 = 1.

p

X.

estimate Fˆ X* * ( x ) is obtained by repeating the
n+h

Fˆε~ ( x) = (n − p ) ∑t = p +11{ε~t ≤ x},
−1

n

εˆ

= (n − p)

steps 5 to 8 B times. The (1-α)% prediction
X n+ h
is
given
by
interval
for

[Q (α / 2), Q (1 − α / 2)],
*

where ε~t = εˆt − εˆ ( • ) and
(•)

(3)

−1

∑

*

where

Q * (.) =

Fˆ X* * (.) are the quantiles of the estimated
n
t = p +1

n+h

εˆt .

bootstrap distribution.

Fortran routines

Module TimeSeriesRoutines
In the module TimeSeriesRoutines are
presented some routines required for the sieve
bootstrap
procedure:
subroutine
AutoCovarianceVector, subroutine YuleWalker,
and subroutine AICCSelection.
SUBROUTINE AutoCovarianceVector(
ACVector, XSeries,MaxLag,Positions)
IMPLICIT NONE
REAL (KIND=8), DIMENSION(0:),
INTENT(OUT) :: ACVector
REAL (KIND=8), DIMENSION(:),
INTENT(IN) :: XSeries
INTEGER, INTENT(IN) :: MaxLag
INTEGER, DIMENSION(:), INTENT(IN),
OPTIONAL :: Positions

This
routine
estimates
the
autocovariances of the XSeries for the orders
from 0 to MaxLag. Notice that the
implementation allows possible missing
observations in the specified Positions. The
expression for the autocovariance estimates is
given by:

γˆ k =

1
n−m

∑

n−k
t =1

wt wt + k ( X t − X )( X t + k − X ),

(6)

where m is the number of missing observations,

X = (n − m) −1 ∑t =1 wt X t and wt is equal 0 if
n

the observation t is missing and otherwise is
equal to 1.
SUBROUTINE YuleWalker(XSeries,
ACMatrix,YWPhi,Residuals)
USE Msimsl
USE Imslf90
IMPLICIT NONE
REAL (KIND=8), DIMENSION(:),
INTENT(IN) :: XSeries
REAL (KIND=8), DIMENSION(:,:),
INTENT(IN) :: ACMatrix
REAL (KIND=8), DIMENSION(:),
INTENT(OUT) :: YWPhi
REAL (KIND=8), DIMENSION(:),
INTENT(OUT) :: Residuals

This routine calculates the Yule-Walker
estimates of the autoregressive coefficient
required in the steps 2 and 7 of sieve bootstrap
procedure. It also calculates the residuals for the
estimated model. The Yule-Walker estimators
can be obtained from the following relation (cf.
Section 8.1 of Brockwell and Davis (1991)):

Γˆ p φˆ p = γˆ p ,

(7)

where Γ̂ p is the estimated autocovariance
matrix

[γˆ ]

p
i − j i , j =1 ,

γˆ p = (γˆ1 , γˆ 2 ,…, γˆ p )' and

φˆ p = (φˆ1 , φˆ2 ,…, φˆp )' is the coefficients vector.
Using (2), the estimated residuals were obtained.
SUBROUTINE AICCSelection(XSeries,
ACVector,PMax,PHat)
IMPLICIT NONE
REAL (KIND=8), DIMENSION(:),
INTENT(IN) :: XSeries
REAL (KIND=8), DIMENSION(0:),
INTENT(IN) :: ACVector
INTEGER, INTENT(IN) :: PMax
INTEGER, INTENT(OUT) :: PHat

This routine implements the AICC
method for selecting the order of the
autoregressive model for XSeries. It considers
models from p = 0 to p = PMax. Instead of using
the subroutine YuleWalker for the different
values of p, it uses the Durbin-Levinson
algorithm (cf. Section 8.2 of Brockwell and
Davis (1991)) which avoids the matrix inversion
required in the direct computation of φ̂ p . The
Durbin-Levinson algorithm uses the following
recursions:

φˆm,m = (γˆm − ∑ j =1φˆm −1, j γˆ m − j ) / vˆm −1 ,

(8)

⎡ φˆm ,1 ⎤ ⎡ φˆm −1,1 ⎤
⎡φˆm −1,m −1 ⎤
⎢ ˆ ⎥ ⎢ ˆ
⎥
⎢ˆ
⎥
⎢ φ m , 2 ⎥ = ⎢ φ m −1, 2 ⎥ − φˆ ⎢φ m −1,m − 2 ⎥,
⎢
⎥ ⎢
⎥ m ,m ⎢
⎥
⎢
⎥ ⎢
⎥
⎢
⎥
⎢⎣φˆm ,m −1 ⎥⎦ ⎢⎣φˆm −1,m −1 ⎥⎦
⎢⎣ φˆm −1,1 ⎥⎦

(9)

m −1

and

vˆm = vˆ m −1 (1 − φˆm2 ,m ),

(10)

with the following initial values: φˆ1,1 = γˆ1 / γˆ0
and vˆ1 = γˆ0 (1 − φˆ12,1 ).
Notice
that
the
subroutine
AICCSelection can be easily modified in order
to use other information criterion as AIC or BIC.
Only the two following sentences required some
minor changes:
MinimumAIC =
RXSize*LOG(ACVector(0)) +
2.0D0*(REAL(I,KIND=8)+1.0D0)
*RXSize/(RXSize - REAL(I+2,KIND=8))
WorkAIC =
RXSize*LOG(VarianceVector(I))
+2.0D0*(REAL(I,KIND=8)+1.0D0)
*RXSize/(RXSize - REAL(I+2,KIND=8))

Routine FESieves
Here are described the subroutine
FESieves which implements the steps 2 to 8 of
the sieve bootstrap procedure. Notice that the
step 1 is implemented by subroutine
AICCSelection.
SUBROUTINE
FESieves(EDF,XSeries,PHat)
USE Msimsl
USE Imslf90
USE TimeSeries
IMPLICIT NONE
REAL (KIND=8), DIMENSION (:,:),
INTENT(OUT) :: EDF
REAL (KIND=8), DIMENSION (:),
INTENT(IN) :: XSeries
INTEGER, INTENT(IN) :: PHat

The inputs of subroutine FESieves are:
the sample XSeries = {X 1 , X 2 , … , X n } and the
selected order, PHat. The output is a MaxLag ×
B matrix, where MaxLag is the maximum
prediction horizon to be considered and B is the
number of resamples.
Step 2 and 7 are implemented by the
following sentences:
CALL YuleWalker(XSeries,
ACMatrix(1:PHat+1, 1:PHat+1),
YWPhi, Residuals)

CALL YuleWalker(WSeries(101:XSize +
100), WACMatrix(1:PHat+1,
1:PHat+1), WYWPhi, YWResiduals)

where the WSeries are the resample obtained
using recursion (4). The estimates YWPhi are
used in recursion (4) and the bootstrap estimates
WYWPhi are used in recursion (5). Also, in the
first call to subroutine YuleWalker, the step 3 is
performed. As mentioned in the previous
section, a bootstrap resample was generated
using (4) with sample size equal to XSize+100
and then discard the first 100 observations by
WSeries(101:XSize + 100).
The resamples of step 5 are obtained by
sampling with replacement from the vector of
centered residuals, WResiduals = WResiduals SUM(WResiduals) / REAL(XSize - PHat,
KIND=8):
DO I = 1, XSize + 100 + MaxLag ! a
resample of centered residual
CALL RNUND(1, XSize-PHat,
RandomIndex)
RResiduals(I)=
WResiduals(RandomIndex)
END DO

Because recursions (4) and (5) are
similar, here, it is only described the prediction
recursion:
DO I = XSize+101, XSize+100+MaxLag
WSeries(I) = RResiduals(I)
DO Ip = 1, PHat
WSeries(I) = WSeries(I) +
WYWPhi(Ip)*WSeries(I-Ip)
END DO
END DO
EDF(1:MaxLag,J)=WSeries(XSize+101:
XSize+100+MaxLag) + XMean

ANDRÉS M. ALONSO

Table 1. Simulation results for Model 1.
Lag
h
1
h
3

Sample size
n
25
50
100
n
25
50
100

Method
Theoretical
Bootstrap
Theoretical
Bootstrap

Coverage (se)
95%
89.03 (0.82)
92.59 (0.52)
93.77 (0.33)
95%
87.50 (0.86)
92.08 (0.49)
93.21 (0.38)

Cov. (below /above)
2.50% / 2.50%
4.44 / 6.53
4.25 / 3.16
3.25 / 2.98
2.50% / 2.50%
5.41 / 7.09
3.97 / 3.95
3.53 / 3.26

Length (se)
3.92
3.74 (0.07)
3.86 (0.05)
3.90 (0.04)
4.92
4.30 (0.08)
4.69 (0.05)
4.77 (0.05)

Table 2. Simulation results for Model 2.
Lag
h
1
h
3

Sample size
n
25
50
100
n
25
50
100

Method
Theoretical
Bootstrap
Theoretical
Bootstrap

Coverage (se)
95%
89.53 (0.85)
92.06 (0.62)
93.31 (0.43)
95%
89.19 (0.79)
91.50 (0.58)
92.49 (0.39)

Cov. (below /above)
2.50% / 2.50%
5.72 / 4.75
3.63 / 4.31
3.49 / 3.20
2.50% / 2.50%
5.15 / 5.66
3.85 / 4.65
3.19 / 4.32

Length (se)
3.93
4.12 (0.08)
3.98 (0.06)
3.96 (0.04)
4.93
4.52 (0.09)
4.62 (0.06)
4.68 (0.05)

Table 3. Simulation results for Model 3.
Lag
h
1
h
3

Sample size
n
25
50
100
n
25
50
100

Method
Theoretical
Bootstrap
Theoretical
Bootstrap

Coverage (se)
95%
89.45 (0.66)
92.44 (0.45)
93.77 (0.36)
95%
89.20 (0.65)
92.79 (0.39)
93.84 (0.34)

Notice that in (5) the bootstrap
autoregressive coefficient is used, WYWPhi,
this allows us to incorporate the parameter
estimation variability in the prediction intervals.
Finally, the α/2 and (1-α/2) quantiles of
the empirical density of forecasts, EDF,
constitutes the prediction interval.

Cov. (below /above)
2.50% / 2.50%
4.73 / 5.82
4.19 / 3.37
3.38 / 2.85
2.50% / 2.50%
4.90 / 5.90
3.68 / 3.53
3.03 / 3.13

Length (se)
3.79
3.54 (0.06)
3.62 (0.04)
3.74 (0.04)
3.93
3.58 (0.06)
3.75 (0.05)
3.88 (0.04)

Results
In this section are briefly described the results of
a simulation experiment using the Fortran
subroutine presented in the previous section. The
following models are used:
•
Model 1: Xt = 0.75 Xt-1 – 0.5 Xt-2 + εt,
where εt are i.i.d. N(0,1).
•
Model 2: Xt = εt – 0.3 εt-1 + 0.7 εt-2,
where εt are i.i.d. N(0,1).

•
Model 3: Xt is a Gaussian process with
autocovariance generating function equal to

G ( z ) =∑k = −∞ γ k z k , where γ k = (| k | +1) −3 .
+∞

The
autoregressive
model
was
considered by Cao et al. (1997), the moving
average model 2 by Pascual et al. (2001) and the
model 3 by Alonso et al. (2002). Notice that
neither model 2 nor model 3 admit a finite AR
representation. Moreover, model 3 does not have
an ARMA representation.
To evaluate the prediction intervals,
their mean coverage and length are used and the
proportions of observations lying out to the left
and to the right of the interval. These quantities
are estimated as follows:
a)
For a combination of model, sample size
and error distribution, simulate a series, and
generate R = 1000 future values Xn+h.
b)
For the bootstrap procedure obtain the
(1-α) prediction interval based on B = 1000
bootstrap resamples.
c)
The coverage is estimated as
C = # Q * (α / 2) ≤ X nr+ h ≤ Q * (1 − α / 2) / R ,

{

}

where X nr+ h with r = 1,2,…,R are the R future
values generated in step a).
In steps a) and b) the “theoretical” and
bootstrap interval lengths are obtained using

LT = X n⎡+Rh(1−α / 2) ⎤ − X n⎡+Rhα / 2 ⎤
and

LB = Q * (1 − α / 2) − Q * (α / 2),
respectively. Finally, the steps a) – c) are
repeated 100 times.
The results are presented in Tables 1 –
3, using three sample sizes n = 25, 50 and 100,
nominal coverage 95% and the prediction lag h
= 1 and 3. Essentially, similar results are
obtained in all cases. Sieve bootstrap performs
reasonably well in all considered models since
the mean coverage and length tend to the
nominal values as the sample size grows. Notice
that for models 2 and 3 the sieve bootstrap never
uses the correct model. The running time for
these three experiments (using a Pentium 4,
running at 2.66GHz) was 22.92, 24.40 and 27.82
seconds, respectively.

Conclusion
It has been shown by Alonso et al. (2002) and
(2003) that, for general linear process, if an AR
approximation that grows with the sample size is
used, it can derive a bootstrap for building
prediction intervals that has the two following
properties: first, the procedure is consistent, that
is, it generates as prediction a random variable
that converges in conditional distribution to the
concerning variable; second, Monte Carlo
simulations show that the proposed procedure
provides better coverage results than previous
methods in general cases. This article describes
a Fortran routine that implement this sieve
bootstrap prediction procedure. Additional
simulation experiments confirm the correct
behavior of the proposed procedure in finite
samples.

Appendix I – Module TimeSeriesRoutines
MODULE TimeSeriesRoutines
REAL (KIND=8), DIMENSION(:), ALLOCATABLE :: ZSeries
INTEGER :: p, d, q, ps, ds, qs, season
CONTAINS
SUBROUTINE AutoCovarianceVector(ACVector,XSeries,MaxLag,Positions)
IMPLICIT NONE
REAL (KIND=8), DIMENSION(0:), INTENT(OUT) :: ACVector
REAL (KIND=8), DIMENSION(:), INTENT(IN) :: XSeries
INTEGER, INTENT(IN) :: MaxLag
INTEGER, DIMENSION(:), INTENT(IN), OPTIONAL :: Positions
! Local variables
INTEGER :: K, I, J, XSize, NMissings
REAL (KIND=8) :: RXSize, XMean
REAL (KIND=8), DIMENSION(:), ALLOCATABLE :: Weights
! First executable statement
XSize = SIZE(XSeries, 1)
ALLOCATE(Weights(XSize))
Weights = 1.0D0
IF (PRESENT(Positions)) THEN
Weights(Positions) = 0.0D0
NMissings = SIZE(Positions, 1)
RXSize = REAL(XSize - NMissings, KIND=8)
ELSE
RXSize = REAL(XSize, KIND=8)
END IF
XMean = SUM(XSeries*Weights)/RXSize
DO K = 0, MaxLag
ACVector(K) = DOT_PRODUCT(&
(XSeries((K+1):XSize) - XMean)*Weights((K+1):XSize), &
(XSeries(1:(XSize-K)) - XMean)*Weights(1:(XSize-K)))/RXSize
END DO
DEALLOCATE(Weights)
END SUBROUTINE AutoCovarianceVector
SUBROUTINE AutoCovarianceMatrix(ACMatrix,XSeries,MaxLag,MSize)
IMPLICIT NONE
REAL (KIND=8), DIMENSION(:,:), INTENT(OUT) :: ACMatrix
REAL (KIND=8), DIMENSION(:), INTENT(IN) :: XSeries
INTEGER, INTENT(IN) :: MaxLag, MSize
! Local variables
INTEGER :: K, I, J, XSize
REAL (KIND=8) :: RXSize, XMean

REAL (KIND=8), DIMENSION(0:MSize) :: ACVector
! First executable statement
XSize = SIZE(XSeries, 1)
RXSize = REAL(XSize, KIND=8)
XMean = SUM(XSeries)/RXSize
DO K = 0, MaxLag+1
ACVector(K) = DOT_PRODUCT(XSeries((K+1):XSize) - XMean, &
XSeries(1:(XSize-K)) - XMean)/RXSize
END DO
DO I = 1, MaxLag+1
DO J = 1, MaxLag+1
ACMatrix(I,J) = ACVector(ABS(I-J))
END DO
END DO
END SUBROUTINE AutoCovarianceMatrix
SUBROUTINE YuleWalker(XSeries,ACMatrix,YWPhi,Residuals)
USE Msimsl
USE Imslf90
IMPLICIT NONE
REAL (KIND=8), DIMENSION(:), INTENT(IN) :: XSeries
REAL (KIND=8), DIMENSION(:,:), INTENT(IN) :: ACMatrix
REAL (KIND=8), DIMENSION(:), INTENT(OUT) :: YWPhi
REAL (KIND=8), DIMENSION(:), INTENT(OUT) :: Residuals
! Local variables
INTEGER :: MSize, XSize, I, J
REAL (KIND=8), DIMENSION(:,:), ALLOCATABLE :: A
REAL (KIND=8), DIMENSION(:), ALLOCATABLE :: B
INTEGER :: M, N, IERR, IOPT, IA, IB
! First executable statement
MSize = SIZE(ACMatrix, 1)
XSize = SIZE(XSeries)
! Initializing LSLDS variables
ALLOCATE(A(MSize-1, MSize-1), B(MSize-1))
A = ACMatrix(1:(MSize-1), 1:(MSize-1))
B = ACMatrix(2:MSize, 1)
M = MSize-1
! Solving the Yule-Walker equations
CALL DLSLDS (M, A, M, B, YWPhi)
! Calculating the YW residuals
Residuals = 0
DO I = (MSize+1), XSize
Residuals(I) = XSeries(I)
DO J = 1, MSize-1
Residuals(I) = Residuals(I) - YWPhi(J)*XSeries(I-J)
END DO
END DO
DEALLOCATE(A, B)
END SUBROUTINE YuleWalker

SUBROUTINE AICCSelection(XSeries,ACVector,PMax,PHat)
IMPLICIT NONE
REAL (KIND=8), DIMENSION(:), INTENT(IN) :: XSeries
REAL (KIND=8), DIMENSION(0:), INTENT(IN) :: ACVector
INTEGER, INTENT(IN) :: PMax
INTEGER, INTENT(OUT) :: PHat
! Local variables
REAL (KIND=8), DIMENSION(:), ALLOCATABLE :: VarianceVector
REAL (KIND=8), DIMENSION(:,:), ALLOCATABLE :: PPhi
REAL (KIND=8) :: VWork, WorkAIC, MinimumAIC, RXSize
INTEGER :: XSize, WorkP, I, J
! First executable statement
ALLOCATE(VarianceVector(PMax))
ALLOCATE(PPhi(PMax, PMax))
XSize = SIZE(XSeries)
RXSize = REAL(XSize, KIND=8)
! Durbin-Levinson Algorithm
PPhi = 0.0D0
PPhi(1, 1) = ACVector(1)/ACVector(0)
VarianceVector(1) = ACVector(0)*(1.0D0 - PPhi(1, 1)**2)
DO I = 2, PMax
VWork = 0
DO J = 1, I-1
VWork = VWork + PPhi(I-1, J)*ACVector(I-J)
ENDDO
PPhi(I, I) = (ACVector(I) - VWork)/VarianceVector(I-1)
DO J = 1, I-1
PPhi(I, J) = PPhi(I-1, J) - PPhi(I, I)*PPhi(I-1, I-J)
ENDDO
VarianceVector(I) = VarianceVector(I-1)*(1.0D0 - PPhi(I, I)**2)
ENDDO
I = 0
MinimumAIC = RXSize*LOG(ACVector(0))+2.0D0*(REAL(I, KIND=8)+1.0D0)* &
RXSize/(RXSize - REAL(I+2, KIND=8))
WorkP = 0
DO I = 1, PMax
WorkAIC = RXSize*LOG(VarianceVector(I))+2.0*(REAL(I, KIND=8) &
+1.0)*RXSize/(RXSize - REAL(I+2, KIND=8))
IF (WorkAIC < MinimumAIC) THEN
MinimumAIC = WorkAIC
WorkP = I
END IF
END DO
PHat = WorkP
DEALLOCATE(PPhi, VarianceVector)
END SUBROUTINE AICCSelection
END MODULE TimeSeriesRoutines

A FORTRAN ROUTINE FOR SIEVE BOOTSTRAP
Appendix II – Routine FESieves
SUBROUTINE FESieves(EDF, XSeries, PHat)
USE Msimsl
USE Imslf90
USE TimeSeries
IMPLICIT NONE
REAL (KIND=8), DIMENSION (:,:), INTENT(OUT) :: EDF
REAL (KIND=8), DIMENSION (:), INTENT(IN) :: XSeries
INTEGER, INTENT(IN) :: PHat
! Local variables
INTEGER :: XSize, MaxLag, B
REAL (KIND=8), DIMENSION(:), ALLOCATABLE :: WSeries
REAL (KIND=8), DIMENSION(:,:), ALLOCATABLE :: ACMatrix
REAL (KIND=8), DIMENSION(:), ALLOCATABLE :: Residuals
REAL (KIND=8), DIMENSION(:), ALLOCATABLE :: YWPhi
REAL (KIND=8) :: XMean
! First executable statement
XSize = SIZE(XSeries, 1)
MaxLag = SIZE(EDF, 1)
B = SIZE(EDF, 2)
ALLOCATE(WSeries(XSize), Residuals(XSize))
ALLOCATE(ACMatrix(PHat+1, PHat+1), YWPhi(PHat))
XMean = SUM(XSeries)/REAL(XSize, KIND=8)
WSeries = XSeries – Xmean
! Steps 2 – 3
CALL AutoCovarianceMatrix(ACMatrix(1:PHat+1, 1:PHat+1), WSeries, &
PHat, PHat+1)
CALL YuleWalker(WSeries, ACMatrix(1:PHat+1, 1:PHat+1), YWPhi, &
Residuals)
! Steps 4 – 8
CALL ESievesBootstrap(EDF,XSeries,YWPhi, Residuals, PHat, MaxLag, B)
DEALLOCATE(ACMatrix, YWPhi, WSeries, Residuals)
CONTAINS
SUBROUTINE ESievesBootstrap(EDF,XSeries,YWPhi,Residuals,PHat,MaxLag,B)
USE Msimsl
USE Imslf90
USE TimeSeries
IMPLICIT NONE
REAL (KIND=8), DIMENSION(:,:), INTENT(OUT) :: EDF
REAL (KIND=8), DIMENSION(:), INTENT(IN) :: XSeries
REAL (KIND=8), DIMENSION(:), INTENT(IN) :: YWPhi
REAL (KIND=8), DIMENSION(:), INTENT(IN) :: Residuals
INTEGER, INTENT(IN) :: PHat, MaxLag, B
! Local variables
INTEGER :: XSize, I, J, Ip, RandomIndex, NOUT, ISEED
REAL (KIND=8), DIMENSION(:), ALLOCATABLE :: WResiduals, RResiduals
REAL (KIND=8), DIMENSION(:), ALLOCATABLE :: WSeries, WYWPhi
REAL (KIND=8), DIMENSION(:,:), ALLOCATABLE :: WACMatrix

REAL (KIND=8), DIMENSION(:), ALLOCATABLE :: YWResiduals
REAL (KIND=8) :: XMean
! First executable statement
XSize = SIZE(XSeries, 1)
ALLOCATE(WSeries(XSize+100+MaxLag))
XMean = SUM(XSeries)/REAL(XSize, KIND=8)
WSeries(1:XSize) = XSeries - XMean
ALLOCATE(WResiduals(XSize - PHat))
WResiduals = Residuals(PHat+1:XSize)
WResiduals = WResiduals - SUM(WResiduals)/REAL(XSize - PHat, KIND=8)
ALLOCATE(RResiduals(XSize+100+MaxLag), WYWPhi(PHat), &
WACMatrix(PHat+1, PHat+1), YWResiduals(XSize))
CALL UMACH (2, NOUT)
CALL RNGET (ISEED)
CALL RNSET (ISEED)
DO J = 1, B
! Steps 4 – 5
DO I = 1, XSize+100+MaxLag
CALL RNUND(1, XSize - PHat, RandomIndex)
RResiduals(I) = WResiduals(RandomIndex)
END DO
! Step 6
WSeries = RResiduals
DO I = PHat+1, XSize+100
DO Ip = 1, PHat
WSeries(I) = WSeries(I) + YWPhi(Ip)*WSeries(I-Ip)
END DO
END DO
! Step 7
CALL AutoCovarianceMatrix(WACMatrix(1:PHat+1, 1:PHat+1), &
WSeries(101:XSize+100), PHat, PHat+1)
CALL YuleWalker(WSeries(101:XSize + 100), &
WACMatrix(1:PHat+1, 1:PHat+1), WYWPhi, YWResiduals)
! Prediction. Step 8
WSeries(101:XSize+100) = XSeries - XMean
DO I = XSize+101, XSize+100+MaxLag
WSeries(I) = RResiduals(I)
DO Ip = 1, PHat
WSeries(I) = WSeries(I) + WYWPhi(Ip)*WSeries(I-Ip)
END DO
END DO
EDF(1:MaxLag, J) = WSeries(XSize+101:XSize+100+MaxLag) + XMean
END DO
DEALLOCATE(WSeries, Residuals, RResiduals, WYWPhi, YWResiduals, &
WACMatrix)
END SUBROUTINE ESievesBootstrap
END SUBROUTINE FESieves

